resonant cavity using scattering transfer matrices

Here I define two functions which convert between S - parameters and T - parameters. These functions only work for 2×2 matrices; I don' t know the general form of the transformation nor whether it really makes sense to use T - parameters for devices with more than two ports (maybe 2N ports?).

Tobin Fricke 2010-08-29

In[1]:=
```
StoT[S_] := {{-Det[S] / S[[2, 1]], S[[1, 1]] / S[[2, 1]]}, {-S[[2, 2]] / S[[2, 1]], 1 / S[[2, 1]]}}
```

In[2]:=
```
TtoS[T_] := {{T[[1, 2]] / T[[2, 2]], Det[T] / T[[2, 2]]}, {1 / T[[2, 2]], -T[[2, 1]] / T[[2, 2]]}}
```

Make sure that the composition of these operations is the identity :

In[3]:=
```
With[{S = Array[s, {2, 2}]}, TtoS[StoT[S]] == S ]
```

Out[3]=
```
True
```

Now I define the S - matrices for a lossless mirror and for free - space :

In[4]:=
```
Smirror = {{r, I t}, {I t, r}}
Sfreespace = Exp[I ϕ] {{0, 1}, {1, 0}}
```

Out[4]=
```
{{r, ⅈ t}, {ⅈ t, r}}
```

Out[5]=
$$\left\{\left\{0, e^{i \phi}\right\}, \left\{e^{i \phi}, 0\right\}\right\}$$

Convert the S - matrices to T - matrices :

In[6]:=
```
Tmirror = StoT[Smirror]
Tfreespace = StoT[Sfreespace]
```

Out[6]=
$$\left\{\left\{-\frac{ⅈ \left(-r^2-t^2\right)}{t}, -\frac{ⅈ r}{t}\right\}, \left\{\frac{ⅈ r}{t}, -\frac{ⅈ}{t}\right\}\right\}$$

Out[7]=
$$\left\{\left\{e^{i \phi}, 0\right\}, \left\{0, e^{-i \phi}\right\}\right\}$$

Define a Fabry - Perot cavity using the T - matrices :

In[8]:=
```
Tmirror1 = Tmirror /. {r → r1, t → t1}; Tmirror2 = Tmirror /. {r → r2, t → t2};
Tfp = Tmirror1 . Tfreespace . Tmirror2;
```

To extract the cavity reflection coefficient, we transform back to the S - matrix :

In[10]:=
```
Sfp = TtoS[Tfp];
```

In[11]:=
```
rc = FullSimplify[Sfp[[1, 1]]];
tc = FullSimplify[Sfp[[2, 1]]];
```

Check whether this is equal to the usual form of rc :

In[13]:=
```
rc == (r1 - (r1^2 + t1^2) r2 Exp[I 2 ϕ]) / (1 - r1 r2 Exp[I 2 ϕ]) // Simplify
```

Out[13]=
```
True
```

It is.